```
> restart():
> # First Computer Lab
> # Math 2270-2 Spring 2012
> # What follows # on that line is a comment.
> # Use comments to put your name and the title of
> # the assignment at the top of the worksheet.
> # Use comments to introduce each problem or major step.
> # End each non-comment line with a semicolon (;) or a colon (:)
> # A semicolon causes BLUE echo and a colon causes no echo.
> #
```

```
> # Hand calculator functions.
> # Use ctrl-Z, backspace, Delete, Arrow keys.
> 2+2; 3*5; 3^6*5/2; 7*(9+11); 10^(1.5);
> sin(Pi/2),cos(Pi),tan(Pi/4);
> #
```

$$4$$
$$15$$
$$\frac{3645}{2}$$
$$140$$
$$31.62277660$$
$$1,\ -1,\ 1 \qquad (1)$$

```
> # Plotting. Right mouse click on the curve to improve the plot.
> plot(sin(x),x=0..Pi);
> #
```




```
> # Printing
> # Use the FILE menu and choose PRINT.
> #
```

```
> # The assignment operator ":=" (colon equals) assigns a value to
  a symbol.
> u:=2; u+3, u*exp(-u^2), sin(Pi*u);
> #
```

$$u := 2$$
$$5,\ 2\,\mathrm{e}^{-4},\ 0 \qquad (2)$$

```
> # Define column vectors with angle brackets.
```

```
> v:=<1,2,3>; w:=<1,0,0>;
> #
```

$$v := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$w := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} \tag{3}$$

```
> #Use a period "." for dot products.
> v.w; v.v;
> #
```

$$1$$

$$14 \tag{4}$$

```
> # Define a matrix using column vectors separated by a bar "|".
> M:=<v|w|<1,1,0>>;
> #
```

$$M := \begin{bmatrix} 1 & 1 & 1 \\ 2 & 0 & 1 \\ 3 & 0 & 0 \end{bmatrix} \tag{5}$$

```
> # Syntax for Matrix multiply and Matrix times Vector.
> M.M; M.v;
> N:=<<1,2,3,4>|<5,6,7,8>|<9,10,11,12>>; N.M;
> #
```

$$\begin{bmatrix} 6 & 1 & 2 \\ 5 & 2 & 2 \\ 3 & 3 & 3 \end{bmatrix}$$

$$\begin{bmatrix} 6 \\ 5 \\ 3 \end{bmatrix}$$

$$N := \begin{bmatrix} 1 & 5 & 9 \\ 2 & 6 & 10 \\ 3 & 7 & 11 \\ 4 & 8 & 12 \end{bmatrix}$$

$$\begin{bmatrix} 38 & 1 & 6 \\ 44 & 2 & 8 \\ 50 & 3 & 10 \\ 56 & 4 & 12 \end{bmatrix} \tag{6}$$

```
> # Error message for incompatible matrix multiply.
> M.N;
> #
```
Error, (in LinearAlgebra:-Multiply) first matrix column dimension (3) <> second matrix row dimension (4)

```
> # Matrices can be entered by rows (preferred).
> P:=Matrix([[1,2,3],[4,5,6],[7,8,9]]);
> #
```

$$P := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix} \tag{7}$$

```
> # Linear combinations of matrices by math-style operations.
> 2*P; P-M; 5*P-3*M;
> #
```

$$\begin{bmatrix} 2 & 4 & 6 \\ 8 & 10 & 12 \\ 14 & 16 & 18 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 1 & 2 \\ 2 & 5 & 5 \\ 4 & 8 & 9 \end{bmatrix}$$

$$\begin{bmatrix} 2 & 7 & 12 \\ 14 & 25 & 27 \\ 26 & 40 & 45 \end{bmatrix} \tag{8}$$

```
> # Use Gaussian Elimination to reduce matrix Q to upper
  triangular form.
> # The percent sign "%" recalls the result of the previous
  computation.
> Q:=<<1,2,1>|<2,3,4>|<1,1,1>>; # Matrix entry by columns
> E1:=<<1,-2,0>|<0,1,0>|<0,0,1>>; # E1:=Matrix([[1,0,0],[-2,1,0],
  [0,0,1]]);
> Q1:=%.Q; # Left multiply by Elimination matrix E1
> E2:=<<1,0,-1>|<0,1,0>|<0,0,1>>; # E2:=Matrix([[1,0,0],[0,1,0],
  [-1,0,1]]);
> Q2:=%.Q1; # Left multiply by Elimination matrix E2
> #
```

$$Q := \begin{bmatrix} 1 & 2 & 1 \\ 2 & 3 & 1 \\ 1 & 4 & 1 \end{bmatrix}$$

$$E1 := \begin{bmatrix} 1 & 0 & 0 \\ -2 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$Q1 := \begin{bmatrix} 1 & 2 & 1 \\ 0 & -1 & -1 \\ 1 & 4 & 1 \end{bmatrix}$$

$$E2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix}$$

$$Q2 := \begin{bmatrix} 1 & 2 & 1 \\ 0 & -1 & -1 \\ 0 & 2 & 0 \end{bmatrix} \quad (9)$$

```
> # The matrix Q has three non-zero pivots, so it is invertible.
> # Find the inverse using two different notations.
> # An answer check is inverse(Q) times Q = identity matrix.
> Q^(-1); 1/Q; %.Q;
> #
```

$$\begin{bmatrix} -\frac{1}{2} & 1 & -\frac{1}{2} \\ -\frac{1}{2} & 0 & \frac{1}{2} \\ \frac{5}{2} & -1 & -\frac{1}{2} \end{bmatrix}$$

$$\begin{bmatrix} -\frac{1}{2} & 1 & -\frac{1}{2} \\ -\frac{1}{2} & 0 & \frac{1}{2} \\ \frac{5}{2} & -1 & -\frac{1}{2} \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (10)$$

```
> # Symbolic computations.
> A:=Matrix([[a[1,1], a[1,2]],[a[2,1],a[2,2]]]); # mouse copy it
> B:=Matrix([[b[1,1], b[1,2]],[b[2,1],b[2,2]]]); # ctrl-K opens a
  line
> C:=Matrix([[c[1,1], c[1,2]],[c[2,1],c[2,2]]]); # ctrl-F is
  find/replace
> #
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \end{bmatrix}$$

$$B := \begin{bmatrix} b_{1,1} & b_{1,2} \\ b_{2,1} & b_{2,2} \end{bmatrix}$$

$$C := \begin{bmatrix} c_{1,1} & c_{1,2} \\ c_{2,1} & c_{2,2} \end{bmatrix} \quad \textbf{(11)}$$

```
> # Verify associativity of matrix multiplication.
> (A.B).C-A.(B.C);
```

$$\Big[\Big[\left(a_{1,1} b_{1,1} + a_{1,2} b_{2,1}\right) c_{1,1} + \left(a_{1,1} b_{1,2} + a_{1,2} b_{2,2}\right) c_{2,1} - a_{1,1}\left(b_{1,1} c_{1,1} + b_{1,2} c_{2,1}\right) - a_{1,2}\left(b_{2,1} c_{1,1} + b_{2,2} c_{2,1}\right), \left(a_{1,1} b_{1,1} + a_{1,2} b_{2,1}\right) c_{1,2} + \left(a_{1,1} b_{1,2} + a_{1,2} b_{2,2}\right) c_{2,2} - a_{1,1}\left(b_{1,1} c_{1,2} + b_{1,2} c_{2,2}\right) - a_{1,2}\left(b_{2,1} c_{1,2} + b_{2,2} c_{2,2}\right)\Big], \Big[\left(a_{2,1} b_{1,1} + a_{2,2} b_{2,1}\right) c_{1,1} + \left(a_{2,1} b_{1,2} + a_{2,2} b_{2,2}\right) c_{2,1} - a_{2,1}\left(b_{1,1} c_{1,1} + b_{1,2} c_{2,1}\right) - a_{2,2}\left(b_{2,1} c_{1,1} + b_{2,2} c_{2,1}\right), \left(a_{2,1} b_{1,1} + a_{2,2} b_{2,1}\right) c_{1,2} + \left(a_{2,1} b_{1,2} + a_{2,2} b_{2,2}\right) c_{2,2} - a_{2,1}\left(b_{1,1} c_{1,2} + b_{1,2} c_{2,2}\right) - a_{2,2}\left(b_{2,1} c_{1,2} + b_{2,2} c_{2,2}\right)\Big]\Big] \quad \textbf{(12)}$$

```
> # The zero matrix is expected. To encourage the maple engine
> # to simplify algebraic expressions, use:
> simplify(%);
> #
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad \textbf{(13)}$$

```
> # Elimination can be done step by step in maple.
> # Load the maple library for linear algebra, as follows.
> # Once per session. The colon removes BLUE printout.
> with(LinearAlgebra):
> # Perform Elimination, showing only the answer, no steps.
> # We choose the system Qx=b, where b:=<1,2,3>:
> b:=<1,2,3>: Q; A1:=<Q|b>;
> GaussianElimination(A1); # ESC key = word completion
> ReducedRowEchelonForm(A1);
> #
```

$$\begin{bmatrix} 1 & 2 & 1 \\ 2 & 3 & 1 \\ 1 & 4 & 1 \end{bmatrix}$$

$$A1 := \begin{bmatrix} 1 & 2 & 1 & 1 \\ 2 & 3 & 1 & 2 \\ 1 & 4 & 1 & 3 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 1 \\ 0 & -1 & -1 & 0 \\ 0 & 0 & -2 & 2 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & -1 \end{bmatrix} \quad (14)$$

```
> # Elimination steps with LinearAlgebra functions. Definitions:
> combo:=(a,s,t,c)->RowOperation(a,[t,s],c);
> swap:=(a,s,t)->RowOperation(a,[t,s]);
> mult:=(a,t,c)->RowOperation(a,t,c);
> A1:=<Q|b>; # Do 9-10 steps with combo, swap, mult.
> A2:=combo(%,1,2,-2); # Invent the other steps.
```

$$combo := (a, s, t, c) \rightarrow LinearAlgebra\text{:-}RowOperation(a, [t, s], c)$$

$$swap := (a, s, t) \rightarrow LinearAlgebra\text{:-}RowOperation(a, [t, s])$$

$$mult := (a, t, c) \rightarrow LinearAlgebra\text{:-}RowOperation(a, t, c)$$

$$A1 := \begin{bmatrix} 1 & 2 & 1 & 1 \\ 2 & 3 & 1 & 2 \\ 1 & 4 & 1 & 3 \end{bmatrix}$$

$$A2 := \begin{bmatrix} 1 & 2 & 1 & 1 \\ 0 & -1 & -1 & 0 \\ 1 & 4 & 1 & 3 \end{bmatrix} \quad (15)$$

```
> # This is a good way to do homework problems. Answer check:
> ReducedRowEchelonForm (A1);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & -1 \end{bmatrix} \quad (16)$$

```
> # There is an interactive Gauss-Jordan Elimination Tutorial
> # in the Student[LinearAlgebra] package. Try it out by
> # un-commenting the next line, then execute the line.
> #Student[LinearAlgebra][GaussJordanEliminationTutor](A1);
> # End of lab1
```